

EJEMPLO INTEGRADOR

Ejemplo 1 – Macro para importar datos de archivos de bloc de notas

Esta macro importa datos de un archivo de bloc de notas en donde se encuentra la facturación de una empresa de un determinado mes y año. La información importada se agrega a una planilla general, previo verificar si existe información duplicada y eliminarla.

La planilla está ubicada en la primera hoja (cuyo nombre es “Facturación”) con el siguiente encabezado en el rango A1:D1

Nro. Factura	Fecha	Producto	Monto
-----------------	-------	----------	-------

La información proveniente del bloc de notas posee estos cuatro datos separados con un punto y coma, como se muestra a continuación:

1;20-01-18;Televisor;40000

El archivo de bloc de notas que contiene la información se denomina según su MES y AÑO, separado con un espacio, como se muestra a continuación:

ENERO 2018

Sub ImportarFacturacion()

' Esta macro importa la facturación proveniente de tablas externas en un archivo de texto (.txt) con un nombre que indique su mes y su año.

Dim MM As String, AA As String 'Declaramos las variables MM = Mes del archivo y AA = Año del archivo

Dim UF As Integer, SF As Integer, NF As Integer 'Declaramos las variables UF = última fila, SF = siguiente fila y NF = nuevas filas

MM = Range("K5").Value 'Asignamos el mes escrito en la celda K5 a la variable MM (que indica el mes del archivo)

AA = Range("K6").Value 'Asignamos el año escrito en la celda K6 a la variable AA (que indica el año del archivo)

UF = WorksheetFunction.CountA(Range("A:A")) 'Asignamos el valor resultante de contar todas las filas en la columna A a la variable UF.

SF = UF + 1 'Asignamos el valor de la siguiente fila sumando el de la última fila (UF) más 1 (que es donde nos posicionaremos para importar los datos)

With ActiveSheet.QueryTables.Add(_

Connection:="TEXT;" & ThisWorkbook.Path & "\VENTAS " & MM & " " & AA & ".txt", _

Destination:=Cells(SF, 1)) 'Crea una consulta que importa los datos de una tabla externa

'("connection" indica la ubicación y nombre del archivo y "destination" su destino en la aplicación de Excel)

.FieldNames = True

.RowNumbers = False

.FillAdjacentFormulas = False

.PreserveFormatting = True

.SaveData = True

.AdjustColumnWidth = False 'Nos permite autoajustar las columnas de la tabla importada (True) o no (False)

.TextFileStartRow = 1

.TextFileParseType = xlDelimited 'Indicamos que para SEPARAR las columnas de la tabla importada usaremos la opción "DELIMITADO"

.TextFileTextQualifier = xlTextQualifierDoubleQuote

.TextFileConsecutiveDelimiter = False

`.TextFileSemicolonDelimiter = True` 'Indicamos que el PUNTO Y COMA (semicolon en inglés) sirve para SEPARAR las columnas de la tabla importada

`.TextFileColumnDataTypes = Array(1, 1, 1, 1)`

`.TextFileTrailingMinusNumbers = True`

`.Refresh BackgroundQuery:=False`

End With

'Borra la consulta (dejando los datos) para que no haya una conexión entre el presente archivo y los de origen

ActiveSheet.QueryTables(1).Delete

'Quitamos los registros duplicados que traería la importación mirando el número de factura (en la primer columna)

Range("A:D").RemoveDuplicates Columns:=1, Header:=xlYes

'Asignamos el valor resultante de restar el conteo de todas las filas en la columna A (que indica la cantidad de filas existentes actuales) y el valor asignado a UF (que indicaba la cantidad de filas existentes anterior a la importación).

NF = WorksheetFunction.CountA(Range("A:A")) - UF

'Indicamos através de un cuadro de mensaje cuál archivo (indicando el mes y el año) y cuántos nuevos registros se importaron

MsgBox "Mes importado: " & LCase(Range("K5").Value) & " " & Range("K6").Value & Chr(10) & "Se importaron " & NF & " filas"

'Dejamos constancia en la celda K8 del último archivo importado y de su fecha y hora de importación

Range("K8").Value = "Última importación: " & Now & " - Mes importado: " & MM & " " & AA

End Sub

Nota: Recordar que podemos grabar la macro que importe los datos (para evitar escribir el código de la importación que suele ser bastante largo y complejo) y luego le podemos hacer los cambios necesarios:

- Indicar que en cada caso queremos elegir un archivo distinto (con un mes y año diferente).
- Indicar que en cada caso queremos que importe la información en una celda distinta (siguiente fila de la tabla).
- Indicar que queremos eliminar el vínculo que se genera entre la información filtrada y la tabla original.
- Quitar los registros duplicados, si se quisieran importar datos que ya son existentes en la tabla.
- Mostrar un cuadro de mensaje que indique el éxito de la importación.